A Simplified Variant of Gödel's Ontological Argument

Christoph Benzmüller
Otto-Friedrich-Universität Bamberg

Preprint; article to appear in *Sophia*.

A Simplified Variant of Gödel's Ontological Argument

Christoph Benzmüller

Abstract A simplified variant of Gödel's ontological argument is presented. The simplified argument is valid already in basic modal logics K or KT, it does not suffer from modal collapse, and it avoids the rather complex predicates of essence (Ess.) and necessary existence (NE) as used by Gödel. The variant presented has been obtained as a side result of a series of theory simplification experiments conducted in interaction with a modern proof assistant system. The starting point for these experiments was the computer encoding of Gödel's argument, and then automated reasoning techniques were systematically applied to arrive at the simplified variant presented. The presented work thus exemplifies a fruitful human-computer interaction in computational metaphysics. Whether the presented result increases or decreases the attractiveness and persuasiveness of the ontological argument is a question I would like to pass on to philosophy and theology.

Keywords Ontological argument · Computational metaphysics · Modal collapse

1 Introduction

Gödel's (1970) ontological argument has attracted significant, albeit controversial, interest among philosophers, logicians and theologians (Sobel, 2004). In this article I present a simplified variant of Gödel's argument that was developed in interaction with the proof assistant system Isabelle/HOL (Nipkow et al., 2002), which is based on classical higher-order logic (Benzmüller & Andrews, 2019). My personal interest in Gödel's argument has been primarily of logical nature. In particular, this interest encompasses the challenge of automating and applying reasoning in quantified modal logics using an universal meta-logical reasoning approach (Benzmüller, 2019) in which (quantified) non-classical logics are semantically embedded in classical higher-order logic. The simplified ontological argument presented below is a side result of this research, which began with a computer encoding of Gödel's argument so that it became amenable to formal analysis and computer-assisted theory simplification experiments; cf. Benzmüller (2020) for more technical details on the most recent series of experiments. The simplified argument selected for presentation in this article has, I believe, the potential to further stimulate the philosophical and theological debate on Gödel's argument, since the simplifications achieved are indeed quite far-reaching:

– Only minimal assumptions about the modal logic used are required. The simplified variant presented is indeed valid in the comparatively weak modal logics K or KT, which only use uncontroversial reasoning principles.[1]

C. Benzmüller
Otto-Friedrich-Universität Bamberg, Kapuzinerstraße 16, 96047 Bamberg, Germany
Freie Universität Berlin, Dep. of Mathematics and Computer Science, Arnimallee 7, 14195 Berlin, Germany
E-mail: christoph.benzmueller@uni-bamberg.de

[1] Some background on modal logic (see also Garson, 2018, and the references therein): The modal operators $\Box$ and $\Diamond$ are employed, in the given context, to capture the alethic modalities "necessarily holds" and "possibly holds", and often the modal logic S5 is used for this. However, logic S5 comes with some rather strong reasoning principles, that could, and have been, be taken as basis for criticism on Gödel's argument. Base modal logic K is comparably uncontroversial, since it only adds the following principles to classical logic: (i) If s is a theorem of K, then so is $\Box s$, and (ii) the distribution axiom $\Box(s \rightarrow t) \rightarrow (\Box s \rightarrow \Box t)$ (if s implies t holds necessarily, then the

– Gödel's argument introduces the comparably complex predicates of essence ($\mathsf{Ess.}$) and necessary existence (NE), where the latter is based on the former. These terms are avoided altogether in the simplified version presented here.
– Above all, a controversial side effect of Gödel's argument, the so-called modal collapse, is avoided. Modal collapse (MC), formally notated as $\forall s\ (s \rightarrow \Box s)$, expresses that "what holds that holds necessarily", which can also be interpreted as "there are no contingent truths" and that "everything is determined". The observation that Gödel's argument implies modal collapse has already been made by Sobel (1987), and Kovač (2012) argues that modal collapse may even have been intended by Gödel. Indeed, the study of modal collapse has been the catalyst for much recent research on the ontological argument. For example, variants of Gödel's argument that avoid modal collapse have been presented by Anderson (1990, 1996) and Fitting (2002), among others, cf. also the formal verification and comparison of these works by Benzmüller and Fuenmayor (2020). In the following, however, it is shown that modal collapse can in fact be avoided by much simpler means.

What I thus present in the remainder is a simple divine theory, derived from Gödel's argument, that does not entail modal collapse.

Since Gödel's (1970) argument was shown to be inconsistent (Benzmüller & Woltzenlogel Paleo, 2016), the actual starting point for the exploration of the simplified ontological argument has been Scott's variant (1972), which is consistent. The terminology and notation used in what follows therefore also remains close to Scott's.

Only one single uninterpreted constant symbol P is used in the argument. This symbol denotes "positive properties", and its meaning is restricted by the postulated axioms, as discussed below. Moreover, the following definitions (or shorthand notations) were introduced by Gödel, respectively Scott:

– An entity x is God-like if it possesses all positive properties.

$$\mathsf{G}(x) \quad \equiv \quad \forall \phi\ (\mathsf{P}(\phi) \rightarrow \phi(x))$$

– A property ϕ is an essence ($\mathsf{Ess.}$) of an entity x if, and only if, (i) ϕ holds for x and (ii) ϕ necessarily entails every property ψ of x (i.e., the property is necessarily minimal).

$$\phi\ \mathsf{Ess.}\ x \quad \equiv \quad \phi(x) \wedge \forall \psi\ (\psi(x) \rightarrow \Box \forall y\ (\phi(y) \rightarrow \psi(y)))$$

Deviating from Gödel, Scott added here the requirement that ϕ must hold for x. Scott found it natural to add this clause, not knowing that it fixed the inconsistency in Gödel's theory, which was discovered by an automated theorem prover (Benzmüller & Woltzenlogel Paleo, 2016). Gödel's (1970) scriptum avoids this conjunct, although it occurred in some of his earlier notes.
– A further shorthand notation, $\mathsf{NE}(x)$, termed necessary existence, was introduced by Gödel. $\mathsf{NE}(x)$ expresses that x necessarily exists if it has an essential property.

$$\mathsf{NE}(x) \quad \equiv \quad \forall \phi\ (\phi\ \mathsf{Ess.}\ x \rightarrow \Box \exists x\ \phi(x))$$

The axioms of Scott's (1972) theory, which constrain the meaning of constant symbol P, and thus also of definition G, are now as follows:

AXIOM 1 Either a property or its negation is positive, but not both.[2]

$$\forall \phi\ (\mathsf{P}(\neg \phi) \leftrightarrow \neg \mathsf{P}(\phi))$$

AXIOM 2 A property is positive if it is necessarily entailed by a positive property.

$$\forall \phi\ \forall \psi\ ((\mathsf{P}(\phi) \wedge (\Box \forall x\ (\phi(x) \rightarrow \psi(x)))) \rightarrow \mathsf{P}(\psi))$$

necessity of s implies the necessity of t). Modal logic KT additionally provides the T axiom: $\Box s \rightarrow s$ (if s holds necessarily, then s), respectively its dual $s \rightarrow \Diamond s$ (if s, then s is possible).
Model logics can be given a possible world semantics, so that $\Box s$ can be read as: for all possible worlds v, which are reachable from a given current world w, we have that s holds in v. And its dual, $\Box s$, thus means: there exists a possible world v, reachable from the current world w, so that s holds in v.

[2] $\neg \phi$ is shorthand for $\lambda x\ \neg \phi(x)$.

AXIOM 3 Being Godlike is a positive property.[3]

$$\mathsf{P}(\mathsf{G})$$

AXIOM 4 Any positive property is necessarily positive (in Scott's words: being a positive property is logical, hence, necessary).

$$\forall\phi\ (\mathsf{P}(\phi) \rightarrow \Box\mathsf{P}(\phi))$$

AXIOM 5 Necessary existence (NE) is a positive property.

$$\mathsf{P}(\mathsf{NE})$$

From this theory the following theorems and corollaries follow; cf. Scott (1972) and Benzmüller and Woltzenlogel Paleo (2014, 2016) for further details. Note that the proofs are valid already in (extensional) modal logic KB, which extends base modal logic K with AXIOM B: $\forall\phi\ (\phi \rightarrow \Box\Diamond\phi)$, or in words, if ϕ then ϕ is necessarily possible.

THEOREM 1 Positive properties are possibly exemplified.

$$\forall\phi\ (\mathsf{P}(\phi) \rightarrow \Diamond\exists x\ \phi(x))$$

Follows from AXIOM 1 and AXIOM 2.
CORO Possibly there exists a God-like being.

$$\Diamond\exists x\ \mathsf{G}(x)$$

Follows from THEOREM 1 and AXIOM 3.
THEOREM 2 Being God-like is an essence of any God-like being.

$$\forall x\ \mathsf{G}(x) \rightarrow \mathsf{G}\ \mathsf{Ess.}\ x$$

Follows from AXIOM 1 and AXIOM 4 using the definitions of Ess. and G.
THEOREM 3 Necessarily, there exists a God-like being.

$$\Box\exists x\ \mathsf{G}(x)$$

Follows from AXIOM 5, CORO, THEOREM2, AXIOM B using the definitions of G and NE.
THEOREM 4 There exists a God-like being.

$$\exists x\ \mathsf{G}(x)$$

Follows from THEOREM 3 together with CORO and AXIOM B.

All claims have been verified with the higher-order proof assistant system Isabelle/HOL (Nipkow et al., 2002) and the sources of these verification experiments are presented in Fig. 2 in the Appendix. This verification work utilised the universal meta-logical reasoning approach (Benzmüller, 2019) in order to obtain a ready to use "implementation" of higher-order modal logic in Isabelle/HOL's classical higher-order logic.

In these experiments only possibilist quantifiers were initially applied and later the results were confirmed for a modified logical setting in which first-order actualist quantifiers for individuals were used, and otherwise possibilist quantifiers. It is also relevant to note that, in agreement with Gödel and Scott, in this article only extensions of (positive) properties paper are considered, in contrast to Fitting (2002), who studied the use of intensions of properties in the context of the ontological argument.

[3] Alternatively, we may postulate A3': The conjunction of any collection of positive properties is positive. Formally, $\forall\mathcal{Z}.(\mathsf{P}os\ \mathcal{Z} \rightarrow \forall X\ (X\bigsqcap\mathcal{Z} \rightarrow \mathsf{P}\ X))$, where $\mathsf{P}os\ \mathcal{Z}$ stands for $\forall X\ (\mathcal{Z}\ X \rightarrow \mathsf{P}\ X)$ and $X\bigsqcap\mathcal{Z}$ is shorthand for $\Box\forall u.(X\ u \leftrightarrow (\forall Y.\ \mathcal{Z}\ Y \rightarrow Y\ u))$.

2 Simplified Variant

Scott's (1972) theory from above has interesting further corollaries, besides modal collapse MC and monotheism (cf. Benzmüller and Woltzenlogel Paleo, 2014, 2016),[4] and such corollaries can be explored using automated theorem proving technology. In particular, the following two statements are implied.

CORO 1 Self-difference is not a positive property.

$$\neg \mathsf{P}(\lambda x\ (x \neq x))$$

Since the setting in this article is extensional, we alternatively get that the empty property, $\lambda x\ \bot$, is not a positive property.

$$\neg \mathsf{P}(\lambda x\ \bot)$$

Both statements follow from AXIOM 1 and AXIOM 2. This is easy to see, because if $\lambda x\ (x \neq x)$ (respectively, $\lambda x\ \bot$) was positive, then, by AXIOM 2, also its complement $\lambda x\ (x = x)$ (respectively, $\lambda x\ \top$) to be so, which contradicts AXIOM 1. Thus, only $\lambda x\ (x = x)$ and $\lambda x\ \top$ can be and indeed are positive, but not their complements.

CORO 2 A property is positive if it is entailed by a positive property.

$$\forall \phi\ \forall \psi\ ((\mathsf{P}(\phi) \wedge (\forall x\ (\phi(x) \rightarrow \psi(x)))) \rightarrow \mathsf{P}(\psi))$$

This follows from AXIOM 1 and THEOREM 4 using the definition of G. Alternatively, the statement can be proved using AXIOM 1, AXIOM B and modal collapse MC.

The above observations are core motivation for our simplified variant of Gödel's argument as presented next; see Benzmüller (2020) for further experiments and explanations on the exploration on this and further simplified variants.

Axioms of the Simplified Ontological Argument

CORO 1 Self-difference is not a positive property.

$$\neg \mathsf{P}(\lambda x\ (x \neq x))$$

(Alternative: The empty property $\lambda x\ \bot$ is not a positive property.)

CORO 2 A property entailed by a positive property is positive.

$$\forall \phi\ \forall \psi\ ((\mathsf{P}(\phi) \wedge (\forall x\ (\phi(x) \rightarrow \psi(x)))) \rightarrow \mathsf{P}(\psi))$$

AXIOM 3 Being Godlike is a positive property.

$$\mathsf{P}(\mathsf{G})$$

As before, an entity x is defined to be God-like if it possesses all positive properties:

$$\mathsf{G}(x) \quad \equiv \quad \forall \phi\ (\mathsf{P}(\phi) \rightarrow \phi(x))$$

From the above axioms of the simplified theory the following successive argumentation steps can be derived in base modal logic K:

LEMMA 1 The existence of a non-exemplified positive property implies that self-difference (or, alternatively, the empty property) is a positive property.

$$(\exists \phi\ (\mathsf{P}(\phi) \wedge \neg \exists x\ \phi(x))) \rightarrow \mathsf{P}(\lambda x\ (x \neq x))$$

This follows from CORO 2, since such a ϕ would entail $\lambda x\ (x \neq x)$.

[4] Monotheism results are of course dependent on the assumed notion of identity. This aspect should be further explored in future work.

$\mathsf{LEMMA\ 2}$ A non-exemplified positive property does not exist.

$$\neg\exists\phi\ (\mathsf{P}(\phi) \land \neg\exists x\ \phi(x))$$

Follows from $\mathsf{CORO\ 1}$ and the contrapositive of $\mathsf{LEMMA\ 1}$.

$\mathsf{LEMMA\ 3}$ Positive properties are exemplified.

$$\forall\phi\ (\mathsf{P}(\phi) \to \exists x\ \phi(x))$$

This is just a reformulation of $\mathsf{LEMMA\ 2}$.

$\mathsf{THEOREM\ 3'}$ There exists a God-like being.

$$\exists x\ \mathsf{G}(x)$$

Follows from $\mathsf{AXIOM\ 3}$ and $\mathsf{LEMMA\ 3}$.

$\mathsf{THEOREM\ 3}$ Necessarily, there exists a God-like being.

$$\Box\exists x\ \mathsf{G}(x)$$

From $\mathsf{THEOREM\ 3'}$ by necessitation.

The model finder *nitpick* Blanchette and Nipkow, 2010 available in Isabelle/HOL can be employed to verify the consistency of this simple divine theory. The smallest satisfying model returned by the model finder consists of one possible world with one God-like entity, and with self-difference, resp. the empty property, not being a positive property. However, the model finder also tell us that it is impossible to prove CORO: $\Diamond\exists x\ \mathsf{G}(x)$, expressing that the existence of a God-like being is possible. The simplest countermodel consists of a single possible world from which no other world is reachable, so that CORO, i.e. $\Diamond\exists x\ \mathsf{G}(x)$, obviously cannot hold for this world, regardless of the truth of $\mathsf{THEOREM\ 3'}$: $\exists x\ \mathsf{G}(x)$ in it. However, the simple transition from the basic modal logic K to the logic KT eliminates this defect. To reach logic KT, $\mathsf{AXIOM\ T}$: $\forall s\ (\Box s \to s)$ is postulated, that is, a property holds if it necessarily holds. This postulate appears uncontroversial. $\mathsf{AXIOM\ T}$ is equivalent to $\mathsf{AXIOM\ T'}$: $\forall s\ (s \to \Diamond s)$, which expresses that a property that holds also possibly holds. Within modal logic KT we can thus obviously prove CORO from $\mathsf{THEOREM\ 3'}$ with the help of $\mathsf{AXIOM\ T'}$.

As an alternative to the above derivation of $\mathsf{THEOREM\ 3}$, we can also proceed in logic KT analogously to the argument given in the introduction.

$\mathsf{THEOREM\ 1}$ Positive properties are possibly exemplified.

$$\forall\phi\ (\mathsf{P}(\phi) \to \Diamond\exists x\ \phi(x))$$

Follows from $\mathsf{CORO\ 1}$, $\mathsf{CORO\ 2}$ and $\mathsf{AXIOM\ T'}$.

CORO Possibly there exists a God-like being.

$$\Diamond\exists x\ \mathsf{G}(x)$$

Follows from $\mathsf{THEOREM\ 1}$ and $\mathsf{AXIOM\ 3}$.

$\mathsf{THEOREM\ 2}$ The possible existence of a God-like being implies its necessary existence.

$$\Diamond\exists x\ \mathsf{G}(x) \to \Box\exists x\ \mathsf{G}(x)$$

Follows from $\mathsf{AXIOM\ 3}$, $\mathsf{CORO\ 1}$ and $\mathsf{CORO\ 2}$.

$\mathsf{THEOREM\ 3}$ Necessarily, there exists a God-like being.

$$\Box\exists x\ \mathsf{G}(x)$$

Follows from CORO and $\mathsf{THEOREM2}$.

$\mathsf{THEOREM\ 3'}$ There exists a God-like being.

$$\exists x\ \mathsf{G}(x)$$

Follows from $\mathsf{THEOREM\ 3}$ with $\mathsf{AXIOM\ T}$.

Interestingly, the above simplified divine theory avoids modal collapse. This is confirmed by the model finder *nitpick*, which reports a countermodel consisting of two possible worlds with one God-like entity.[5]

The above statements were all formally verified with Isabelle/HOL. As with Scott's variant, only possibilist quantifiers were used initially, and later the results were confirmed also for a modified logical setting in which first-order actualist quantifiers for individuals were used, and possibilist quantifiers otherwise. The Isabelle/HOL sources of the conducted verification studies are presented in Figs. 1-4 in the Appendix.

In the related exploratory studies (Benzmüller, 2020), a suitably adapted notion of a modal ultrafilter was additionally used to support the comparative analysis of different variants of Gödel's ontological argument, including those proposed by Anderson and Gettings (1996) and Fitting (2002), which avoid modal collapse. These experiments are a good demonstration of the maturity that modern theorem proving systems have reached. These systems are ready to fruitfully support the exploration of metaphysical theories.

The development of Gödel's ontological argument has recently been addressed by Kanckos and Lethen (2019). They discovered previously unknown variants of the argument in Gödel's Nachlass, whose relation to the presented simplified variants should be further investigated in future work. The version No. 2 they reported has meanwhile been formalised and verified in Isabelle/HOL, similar to the work presented above. This version No. 2 avoids the notions of essence and necessary existence and associated definitions/axioms, just as our simplified version does. However, this version, in many respects, also differs from ours, and it assumes a higher-modal modal logic S5.

3 Discussion

Whether the simplified variant of Gödel's ontological argument presented in this paper actually increases or decreases the argument's appeal and persuasiveness is a question I would like to pass on to philosophy and theology. As a logician, I see my role primarily as providing useful input and clarity to promote informed debate.

I have shown how a significantly simplified version of Gödel's ontological variant can be explored and verified in interaction with modern theorem proving technology. Most importantly, this simplified variant avoids modal collapse, and some further issues, which have triggered criticism on Gödel's argument in the past. Future work could investigate the extent to which such theory simplification studies could even be fully automated. The resulting rational reconstructions of argument variants would be very useful in gaining more intuition and understanding of the theory in question, in this case a theistic theory, which in turn could lead to its demystification and also to the identification of flawed discussions in the existing literature.

In future work, I would like to further deepen ongoing studies of Fitting's (2002) proposal, which works with intensions rather than extension of (positive) properties.

Acknowledgements: I thank Andrea Vestrucci for valuable comments that helped improve this article.

References

Anderson, C. A. (1990). Some emendations of Gödel's ontological proof. *Faith and Philosophy*, *7*(3), 291–303.

Anderson, C. A., & Gettings, M. (1996). Gödel's ontological proof revisited. *Gödel'96: Logical Foundations of Mathematics, Computer Science, and Physics: Lecture Notes in Logic 6* (pp. 167–172). Springer.

Benzmüller, C. (2019). Universal (meta-)logical reasoning: Recent successes. *Science of Computer Programming*, *172*, 48–62. https://doi.org/10.1016/j.scico.2018.10.008

Benzmüller, C. (2020). A (Simplified) Supreme Being Necessarily Exists, says the Computer: Computationally Explored Variants of Gödel's Ontological Argument. *Proceedings of the 17th International Conference on Principles of Knowledge Representation and Reasoning, KR 2020*, 779–789. https://doi.org/10.24963/kr.2020/80

Benzmüller, C., & Andrews, P. (2019). Church's type theory. In E. N. Zalta (Ed.), *The stanford encyclopedia of philosophy* (Summer 2019). Metaphysics Research Lab, Stanford University.

[5] In this countermodel, the possible worlds $i1$ and $i2$ are reachable from $i2$, but only world $i1$ can be reached from $i1$. Moreover, there is non-positive property ϕ which holds for e in world $i2$ but not in $i1$. Apparently, in world $i2$, modal collapse $\forall s\ (s \rightarrow \Box s)$ is not validated. The positive properties include $\lambda x\ \top$.

Benzmüller, C., & Fuenmayor, D. (2020). Computer-supported analysis of positive properties, ultrafilters and modal collapse in variants of Gödel's ontological argument. *Bulletin of the Section of Logic*, *49*(2), 127–148. https://doi.org/10.18778/0138-0680.2020.08

Benzmüller, C., & Woltzenlogel Paleo, B. (2014). Automating Gödel's ontological proof of God's existence with higher-order automated theorem provers. *ECAI 2014*, *263*, 93–98. https://doi.org/10.3233/978-1-61499-419-0-93

Benzmüller, C., & Woltzenlogel Paleo, B. (2016). The inconsistency in Gödel's ontological argument: A success story for AI in metaphysics. In S. Kambhampati (Ed.), *IJCAI 2016* (pp. 936–942). AAAI Press.

Blanchette, J. C., & Nipkow, T. (2010). Nitpick: A counterexample generator for higher-order logic based on a relational model finder. In M. Kaufmann & L. C. Paulson (Eds.), *Interactive theorem proving — ITP 2010* (pp. 131–146). Springer.

Fitting, M. (2002). *Types, tableaus, and Gödel's god*. Kluwer.

Garson, J. (2018). Modal logic. In E. N. Zalta (Ed.), *The stanford encyclopedia of philosophy* (Fall 2018). Metaphysics Research Lab, Stanford University.

Gödel, K. (1970). Appendix A. Notes in Kurt Gödel's Hand. In J. Sobel (Ed.), *Logic and theism: Arguments for and against beliefs in god* (pp. 144–145). Cambridge University Press.

Kanckos, A., & Lethen, T. (2019). The development of Gödel's ontological proof. *The Review of Symbolic Logic*. https://doi.org/10.1017/S1755020319000479

Kovač, S. (2012). Modal collapse in Gödel's ontological proof. In M. Szatkowski (Ed.), *Ontological proofs today* (pp. 50–323). Ontos Verlag.

Nipkow, T., Paulson, L. C., & Wenzel, M. (2002). *Isabelle/HOL — a proof assistant for higher-order logic* (Vol. 2283). Springer.

Scott, D. S. (1972). Appendix B: Notes in Dana Scott's Hand. In J. Sobel (Ed.), *Logic and theism: Arguments for and against beliefs in god* (pp. 145–146). Cambridge University Press.

Sobel, J. H. (1987). Gödel's ontological proof. In J. J. Tomson (Ed.), *On Being and Saying. Essays for Richard Cartwright* (pp. 241–261). MIT Press.

Sobel, J. H. (2004). *Logic and theism: Arguments for and against beliefs in god*. Cambridge University Press.

Appendix: Sources of Conducted Experiments

```
theory HOML imports Main  (*C. Benzmüller, 2021*)
begin
(*unimportant*) declare [[smt_solver=cvc4,smt_oracle]]
(*unimportant*) declare[[syntax_ambiguity_warning=false]]
(*unimportant*) nitpick_params[user_axioms,card=2]
(*Type declarations and type abbreviations*)
typedecl i (*Possible worlds*)
typedecl e (*Individuals*)
type_synonym σ = "i⇒bool" (*World-lifted propositions*)
type_synonym γ = "e⇒σ" (*Lifted predicates*)
type_synonym μ = "σ⇒σ" (*Unary modal connectives*)
type_synonym ν = "σ⇒σ⇒σ" (*Binary modal connectives*)

(*Modal logic connectives (operating on truth-sets)*)
abbreviation c1::σ ("⊥") where "⊥ ≡ λw. False"
abbreviation c2::σ ("⊤") where "⊤ ≡ λw. True"
abbreviation c3::μ ("¬_"[52]53) where "¬φ ≡ λw.¬(φ w)"
abbreviation c4::ν (infix"∧"50) where "φ∧ψ ≡ λw.(φ w)∧(ψ w)"
abbreviation c5::ν (infix"∨"49) where "φ∨ψ ≡ λw.(φ w)∨(ψ w)"
abbreviation c6::ν (infix"→"48) where "φ→ψ ≡ λw.(φ w)⟶(ψ w)"
abbreviation c7::ν (infix"↔"47) where "φ↔ψ ≡ λw.(φ w)⟷(ψ w)"
consts R::"i⇒i⇒bool" ("_r_")  (*Accessibility relation*)
abbreviation c8::μ ("□_"[54]55) where "□φ ≡ λw.∀v.(wrv)⟶(φ v)"
abbreviation c9::μ ("◇_"[54]55) where "◇φ ≡ λw.∃v.(wrv)∧(φ v)"
abbreviation c10::"γ⇒γ" ("¬_"[52]53) where "¬Φ ≡ λx.λw.¬(Φ x w)"
abbreviation c11::"e⇒e⇒σ" ("_=_") where "x=y ≡ λw.(x=y)"
abbreviation c12::"e⇒e⇒σ" ("_≠_") where "x≠y ≡ λw.(x≠y)"

(*Polymorphic possibilist quantification*)
abbreviation q1::"('a⇒σ)⇒σ" ("∀") where "∀Φ ≡ λw.∀x.(Φ x w)"
abbreviation q2 (binder"∀"[10]11) where "∀x. φ(x) ≡ ∀φ"
abbreviation q3::"('a⇒σ)⇒σ" ("∃") where "∃Φ ≡ λw.∃x.(Φ x w)"
abbreviation q4 (binder"∃"[10]11) where "∃x. φ(x) ≡ ∃φ"

(*Actualist quantification for individuals*)
consts existsAt::γ ("_@_")
abbreviation q5::"γ⇒σ" ("∀ᴱ") where "∀ᴱΦ ≡ λw.∀x.(x@w)⟶(Φ x w)"
abbreviation q6 (binder"∀ᴱ"[8]9) where "∀ᴱx. φ(x) ≡ ∀ᴱφ"
abbreviation q7::"γ⇒σ" ("∃ᴱ") where "∃ᴱΦ ≡ λw.∃x.(x@w)∧(Φ x w)"
abbreviation q8 (binder"∃ᴱ"[8]9) where "∃ᴱx. φ(x) ≡ ∃ᴱφ"

(*Meta-logical predicate for global validity*)
abbreviation g1::"σ⇒bool" ("⌊_⌋") where "⌊ψ⌋ ≡  ∀w. ψ w"
end
```

Figure 1 The universal meta-logical reasoning approach at work: exemplary shallow semantic embedding of modal higher-order logic K in classical higher-order logic.

```
theory ScottVariant imports HOML          (*C. Benzmüller, 2021*)
begin
(*Positive properties*)
consts posProp::"γ⇒σ" ("𝒫")

(*Definition of God-like*)
definition G ("𝒢") where "𝒢(x) ≡ ∀Φ.(𝒫(Φ) ⟶ Φ(x))"

(*Definitions of Essence and Necessary Existence*)
definition Ess ("_Ess._"[80,80]81)
  where "Φ Ess. x ≡ Φ(x) ∧ (∀Ψ. Ψ(x) ⟶ □(∀y. Φ(y) ⟶ Ψ(y)))"
definition NE ("𝒩ℰ") where "𝒩ℰ(x) ≡ ∀Φ. Φ Ess. x ⟶ □(∃x. (Φ(x)))"

(*Axioms of Scott's variant*)
axiomatization where
 AXIOM1: "⌊∀Φ. 𝒫(¬Φ) ↔ ¬𝒫(Φ)⌋" and
 AXIOM2: "⌊∀Φ Ψ. 𝒫(Φ) ∧ □(∀x. Φ(x) ⟶ Ψ(x)) ⟶ 𝒫(Ψ)⌋" and
 AXIOM3: "⌊𝒫(𝒢)⌋" and
 AXIOM4: "⌊∀Φ. 𝒫(Φ) ⟶ □𝒫(Φ)⌋" and
 AXIOM5: "⌊𝒫(𝒩ℰ)⌋"

(*We only need the B axiom*)
axiomatization where B:  "⌊∀Φ. Φ ⟶ □◇Φ⌋" (*Logic KB*)
lemma B': "∀x y. ¬xry ∨ yrx" using B by fastforce

(*Verifying Scott's proof from*)
theorem THEOREM1: "⌊∀Φ. 𝒫(Φ) ⟶ ◇(∃x. Φ(x))⌋"
  using AXIOM1 AXIOM2 by blast
theorem CORO: "⌊◇(∃x. 𝒢(x))⌋" using THEOREM1 AXIOM3 by simp
theorem THEOREM2: "⌊∀x. 𝒢(x)⟶ 𝒢 Ess. x⌋"
  by (smt AXIOM1 AXIOM4 Ess_def G_def)
theorem THEOREM3: "⌊□(∃x. 𝒢(x))⌋"
  by (metis AXIOM5 B' CORO G_def NE_def THEOREM2)

theorem THEOREM4: "⌊∃x. 𝒢(x)⌋" using B' CORO THEOREM3 by blast

lemma C1: "⌊¬𝒫(λx. x≠x)⌋" using AXIOM1 AXIOM2 by blast
lemma C1': "⌊¬𝒫(λx. ⊥)⌋" using AXIOM1 AXIOM2 by blast
lemma C2: "⌊∀Φ Ψ. 𝒫(Φ) ∧ (∀x. Φ(x) ⟶ Ψ(x)) ⟶ 𝒫(Ψ)⌋"
   by (metis AXIOM1 G_def THEOREM4)

(*Consistency*)
lemma True nitpick[satisfy] oops (*Model found*)
end
```

Figure 2 Verification of Scott's variant of Gödel's ontological argument in modal higher-order logic KB, using first-order and higher -order possibilistic quantifiers; the theory HOML from Fig. 1 is imported.

```
theory SimplifiedOntologicalArgument      (*C. Benzmüller, 2021*)
  imports HOML
begin
(*Positive properties*)
consts posProp::"γ⇒σ" ("𝒫")

(*Definition of Godlike*)
definition G ("𝒢") where "𝒢(x) ≡ ∀Φ.(𝒫(Φ) → Φ(x))"

(*Axiom's of new variant based on ultrafilters*)
axiomatization where
 CORO1: "⌊¬(𝒫(λx.(x≠x)))⌋" and
 CORO2: "⌊∀Φ Ψ. 𝒫(Φ) ∧ (∀x. Φ(x) → Ψ(x)) → 𝒫(Ψ)⌋" and
 AXIOM3: "⌊𝒫 𝒢⌋"

(*Verifying the simplified ontological argument; version 1*)
lemma LEMMA1: "⌊(∃Φ.(𝒫(Φ) ∧ ¬(∃x. Φ(x)))) → 𝒫(λx.(x≠x))⌋"
  by (meson CORO2)
lemma LEMMA2: "⌊¬(∃Φ.(𝒫(Φ) ∧ ¬(∃x. Φ(x))))⌋"
  using CORO1 LEMMA1 by blast
lemma LEMMA3: "⌊∀Φ.(𝒫(Φ) → (∃x. Φ(x)))⌋" using LEMMA2 by blast
theorem THEOREM3': "⌊∃x. 𝒢(x)⌋" using AXIOM3 LEMMA3 by auto
theorem THEOREM3: "⌊□(∃x. 𝒢(x))⌋" by (simp add: THEOREM3')
theorem CORO: "⌊◇(∃x. 𝒢(x))⌋" nitpick oops (*countermodel*)

(*Modal Logic T*)
axiomatization where T: "⌊∀φ. □φ → φ⌋"
lemma T': "⌊∀φ. φ → ◇φ⌋" by (metis T)

(*Verifying the simplified ontological argument; version 2*)
theorem THEOREM1: "⌊∀Φ. 𝒫(Φ) → ◇(∃x. Φ(x))⌋"
  by (metis CORO1 CORO2 T')
theorem CORO: "⌊◇(∃x. 𝒢(x))⌋" using AXIOM3 THEOREM1 by auto
theorem THEOREM2: "⌊◇(∃x. 𝒢(x)) → □(∃x. 𝒢(x))⌋"
  by (meson AXIOM3 CORO1 CORO2)
theorem THEO3: "⌊□(∃x. 𝒢(x))⌋" using CORO THEOREM2 by blast
theorem THEO3': "⌊∃x. 𝒢(x)⌋" by (metis T THEO3)

lemma MC: "⌊∀Φ. Φ → □Φ⌋" nitpick oops (*countermodel*)

(*Consistency*)
lemma True nitpick[satisfy] oops (*Model found*)
end
```

Figure 3 Simplified ontological argument in modal logic K, respectively KT, using possibilist first-order and higher-order quantifiers.

```
theory SimplifiedOntologicalArgumentActualist (*C. Benzmüller, 2021*)
  imports HOML
begin
(*Positive properties*)
consts posProp::"γ⇒σ" ("𝒫")

(*Definition of Godlike*)
definition G ("𝒢") where "𝒢(x) ≡ ∀Φ.(𝒫(Φ) → Φ(x))"

(*Axiom's of new variant based on ultrafilters*)
axiomatization where
 CORO1: "⌊¬(𝒫(λx.(x≠x)))⌋" and
 CORO2: "⌊∀Φ Ψ. 𝒫(Φ) ∧ (∀ᴱx. Φ(x) → Ψ(x)) → 𝒫(Ψ)⌋" and
 AXIOM3: "⌊𝒫 𝒢⌋"

(*Verifying the simplified ontological argument; version 1*)
lemma LEMMA1: "⌊(∃Φ.(𝒫(Φ) ∧ ¬(∃ᴱx. Φ(x)))) → 𝒫(λx.(x≠x))⌋"
  by (meson CORO2)
lemma LEMMA2: "⌊¬(∃Φ.(𝒫(Φ) ∧ ¬(∃ᴱx. Φ(x))))⌋"
  using CORO1 LEMMA1 by blast
lemma LEMMA3: "⌊∀Φ.(𝒫(Φ) → (∃ᴱx. Φ(x)))⌋" using LEMMA2 by blast
theorem THEOREM3': "⌊∃ᴱx. 𝒢(x)⌋" using AXIOM3 LEMMA2 by presburger
theorem THEOREM3: "⌊□(∃ᴱx. 𝒢(x))⌋" by (simp add: THEOREM3')
theorem CORO: "⌊◇(∃x. 𝒢(x))⌋" nitpick oops (*countermodel*)

(*Modal Logic T*)
axiomatization where T: "⌊∀φ. □φ → φ⌋"
lemma T': "⌊∀φ. φ → ◇φ⌋" by (metis T)

(*Verifying the simplified ontological argument; version 2*)
theorem THEOREM1: "⌊∀Φ. 𝒫(Φ) → ◇(∃ᴱx. Φ(x))⌋"
  by (metis CORO1 CORO2 T')
theorem CORO: "⌊◇(∃ᴱx. 𝒢(x))⌋" using AXIOM3 THEOREM1 by auto
theorem THEOREM2: "⌊◇(∃ᴱx. 𝒢(x)) → □(∃ᴱx. 𝒢(x))⌋"
  by (meson AXIOM3 CORO1 CORO2)
theorem THEO3: "⌊□(∃ᴱx. 𝒢(x))⌋" using CORO THEOREM2 by blast
theorem THEO3': "⌊∃ᴱx. 𝒢(x)⌋" by (metis T THEO3)

lemma MC: "⌊∀Φ. Φ → □Φ⌋" nitpick oops (*countermodel*)

(*Consistency*)
lemma True nitpick[satisfy] oops (*Model found*)
end
```

Figure 4 Simplified ontological argument in modal logic K, respectively KT, using actualist quantifiers first-order quantifiers and possibilist higher-order quantifiers.