Cantor's Theorem without Reductio Ad Absurdum

Christoph Benzmüller and David Fuenmayor

February 9, 2021

Abstract

Reductio ad absurdum can easily be avoided in the proof of Cantor's theorem.

1 The surjective Cantor theorem

Cantor's theorem, an important result in set theory, states that the cardinality of a set is strictly less than the cardinality of its powerset. It is amongst the most well known theorems in maths, listed occasionally even in some (not entirely uncontroversial) theorems chart lists [2].

The surjective variant of Cantor's theorem states that *"there exists no surjective map from a set to its powerset"*. In classical higher-order logic (HOL) [5] this statement can be elegantly encoded as

$$\neg\exists f_{i\to(i\to o)}\forall p_{i\to o}\exists x_i\,(f\ x = p)$$

where the type i denotes an arbitrary non-empty set and, correspondingly, type $i \to o$ is associated with the powerset of that set (sets can be identified in HOL with their characteristic functions).

In the area of higher-order automated theorem proving the surjective Cantor theorem has, since about four decades, served as an entry level challenge problem, and it was the TPS prover [3] of Peter Andrews and his students that automated the proof for the first time [4].

The numerous human constructed proofs to be found in the literature, in online repositories and in the libraries of modern proof assistant systems (see e.g. [1]) combine the construction of a diagonalization set with a *reductio ad absurdum* (proof by contradiction) proof argument; see Sect. 2. However, as we demonstrate in Sect. 3, the use reductio ad absurdum can easily be avoided. Since reductio ad absurdum is sometimes even debated as a potential weakness of Cantor's proof argument, we have decided to present this alternative proof here. Note it may also be useful for educational purposes.

2 Traditional proof

A traditional proof of the surjective Cantor theorem is as follows:

Assume there is a surjective f. Consider the following choice for p (the diagonal set) $p := \{x \mid x \notin (f\ x)\}$. There must be an a such that $(f\ a) = p$ (since f is surjective by assumption). But then we have $a \in (f\ a)$ iff $a \in p$ iff $a \notin (f\ a)$ (by definition of p). Contradiction.

To formalize the statement and this proof in Isabelle/HOL we start out with importing the theory Main of Isabelle/HOL, and we declare an arbitrary base type `i`.

theory *CantorsTheorem* **imports** *Main*
begin

typedecl *i*

The theorem and the proof (in pure HOL) are then encoded as follows (see also [6, Sect.4.1] and [7, Sect. 5.1]):

lemma *CantorSurjective*: $\neg(\exists f{::}i{\Rightarrow}i{\Rightarrow}bool.\forall p.\exists x.\ f\ x = p)$
proof
  assume $\exists f{::}i{\Rightarrow}i{\Rightarrow}bool.\forall p.\exists x.\ f\ x = p$
  then obtain $f :: i{\Rightarrow}i{\Rightarrow}bool$
    where *1*: $\forall p.\exists x.\ f\ x = p$ **by** *blast*
  let $?P = \lambda x.\ \neg\ f\ x\ x$
  have $\exists x.\ ?P = f\ x$ **using** *1* **by** *metis*
  then obtain $a{::}i$
    where $?P = f\ a$ **by** *blast*
  then have $?P\ a \longleftrightarrow f\ a\ a$ **by** *metis*
  then have $\neg\ f\ a\ a \longleftrightarrow f\ a\ a$ **by** *blast*
  then show *False* **by** *blast*
qed

3 Alternative proof

The alternative proof argument, which avoids reductio ad absurdum, goes as follows:

Fix a map f from a set to its powerset. Obviously, for all w there is a v such that $v \notin (f\ w)$ iff $v \notin (f\ v)$ (simply choose $v = w$). Now, choose $p := \{x \mid x \notin (f\ x)\}$ (the diagonal set). Then, by definition, for all w there is a v with $v \notin (f\ w)$ iff $v \in p$. Hence there is such a p. By Boolean and functional extensionality, after pulling negation outwards, it follows: there exists p so that $(f\ w) \neq p$ for all w. f has been chosen arbitrary, thus the latter is true for all f. Pulling negation outwards gives us the theorem.

In Isabelle/HOL this proof can be encoded as follows:

lemma *CantorSurjective'*: $\neg(\exists f{::}i{\Rightarrow}i{\Rightarrow}bool.\forall p.\exists x.\ f\ x = p)$
proof –
  {fix $F{::}i{\Rightarrow}i{\Rightarrow}bool$
    have $\forall w.\ \exists v.\ \neg(F\ w\ v) \longleftrightarrow \neg(F\ v\ v)$ **by** *auto* — choose v=w
    hence *let* $P = (\lambda x.\ \neg F\ x\ x)$ *in* $\forall w.\ \exists v.\ \neg(F\ w\ v) \longleftrightarrow (P\ v)$
      by *simp* — lambda-conversion and replacement
    hence $\exists p.\ \forall w.\ \exists v.\ \neg(F\ w\ v) \longleftrightarrow (p\ v)$ **by** *auto* — exists-introduction
    hence $\exists p.\ \forall w.\ \neg(\forall v.\ (F\ w\ v) \longleftrightarrow (p\ v))$ **by** *auto* — pull negation outwards
    hence $\exists p.\ \forall w.\ \neg(F\ w = p)$ **by** *metis* — Boolean and functional extensionality
  }
  hence $\forall f{::}i{\Rightarrow}i{\Rightarrow}bool.\ \exists p.\ \forall w.\ \neg(f\ w = p)$ **by** *simp* — F was chosen arbitrary
  hence $\neg(\exists f{::}i{\Rightarrow}i{\Rightarrow}bool.\forall p.\exists x.\ f\ x = p)$ **by** *simp* — pull negation outwards
  thus *?thesis* **.** — we are done
qed

end

References

[1] Theory Cantor (in Isabelle/HOL). https://isabelle.in.tum.de/library/HOL/HOL-Isar_Examples/Cantor.html. Accessed Feb 5, 2021.

[2] The top 100 theorems in Isabelle. http://www.cse.unsw.edu.au/~kleing/top100/. See also http://www.cs.ru.nl/~freek/100/.

[3] P. B. Andrews, M. Bishop, S. Issar, D. Nesmith, F. Pfenning, and H. Xi. TPS: A theorem-proving system for classical type theory. *J. Autom. Reason.*, 16(3):321–353, 1996.

[4] P. B. Andrews, E. Longini-Cohen, D. Miller, and F. Pfenning. Automating higher order logics. *Contemp. Math*, 29:169–192, 1984.

[5] C. Benzmüller and P. Andrews. Church's type theory. In E. N. Zalta, editor, *The Stanford Encyclopedia of Philosophy*. Metaphysics Research Lab, Stanford University, summer 2019 edition, 2019.

[6] T. Nipkow. Programming and proving in Isabelle/HOL. https://isabelle.in.tum.de/dist/doc/prog-prove.pdf, 2020.

[7] T. Nipkow and G. Klein. *Concrete Semantics - With Isabelle/HOL*. Springer, 2014.